
TITLE: 
Cavi51: steam explosion in a 3D cavity. 
PROBLEM: 
This problem was suggested by ISMES (I) in the mid-nineties as a check of the code capability to 
model an extremely violent phenomenon, the steam explosion caused by sudden drop of hot corium 
material (following an HCDA) into a pool of liquid water situated in a cavity between the primary 
reactor vessel (supposed to be unperforated!) and the rigid cylindrical walls. The problem is 3D 
because of a lateral corridor that opens in the cavity and puts it in communication with an external 
room. All walls are considered rigid, but the deformations (plasticity) of the primary vessel must be 
modeled. 
MESH: 
The model is 3D and uses 7664 elements FL38 for the fluid and 3072 shell elements COQI for the 
vessel. The calculation is ALE and starts with a hot and high-pressure steam bubble at the center of the 
liquid pool. 
MATERIALS: 
Because of the huge distortions and the thinness of the cavity the only way to simulate this test is by 
means of the multi-phase multi-component fluid material model (FLMP). This consists of three 
components: the liquid water (WT0 model) for the pool, a perfect gas for the air and another perfect 
gas, initially at high pressure and temperature, for the steam bubble. The structure uses the VM23 
elasto-plastic material. 
BOUNDARY CONDITIONS: 
The vessel is entirely blocked along the top circle. The fluid-structure interaction is automaticcaly 
modelled by the FSA directive along the fluid-vessel interface, and by FSR along the external surface 
of the fluid (the cavity walls are supposed rigid). 
LOADING: 
The system is initially at rest, but not in equilibrium. 
CALCULATION: 
The calculation is performed up to 50 ms. At the final time, the vessel has undergone large 
plastifications at the bottom and also near the top (water-hammer effect), and the liquid water has 
invaded bothe the cavity and the external room. 
RESULTS: 
Although no comparison with experiments can be done on this problem, the results appear physically 
correct. 

POST-TREATMENT 
Several animations of the computed results from this calculation are available on the EUROPLEXUS 
Consortium Web site. 
 
REFERENCES: 
This calculation is detailed in: 
F. Casadei: "Generalization of the Finite Element Model for Compressible Fluids in PLEXIS-3C to 
Multi-Phase Flows." Technical Note N. I.97.33, March 1997. (available on the EUROPLEXUS 
Consortium Web site). 
 
Numerical Solution 
 
CAVI51 
The mesh generation file (K2000) is: 
 
*%siz 300 
 
opti echo 0; 
opti donn 'D:\Users\Folco\plexis3c\proc\pxordpoi.proc'; 
opti donn 'D:\Users\Folco\plexis3c\proc\pxrota3d.proc'; 
opti donn 'D:\Users\Folco\plexis3c\proc\pxvolu3d.proc'; 
opti donn 'D:\Users\Folco\plexis3c\proc\px4to3.proc'; 
opti echo 1; 
opti trac psc ftra 'vl_jrc_cavi51_mesh.ps'; 
* 
opti titr 'CAVI - 51'; 
opti dime 3 elem cub8; 
* 
p0=0 0 0; 
p1=2.35 0 0; 
p2=2.35 0 11; 
p3=2.20 0 11;p3p=p3 plus p0; 
p4=2.35 0 3.9; 
p5=2.20 0 3.9;p5p=p5 plus p0; 
p6=0 0 1.7;p6p=p6 plus p0; 
p7=0 0 1.45; 
p8=0.6 0 0.85; 
pcb=0 0 0.85; 
pcv=0 0 3.9; 
p9=0 0 0.25; 
p10=0.709882641 0 1.817677586; 
p10p=p10 plus p0; 
p11=0.75 0 1.7; 
p12=2.35 0 1.7;p12b=2.35 0 2.0; 
p13p=p5p plus (0 0 0.39444); 
p14=2.35 0 2.9571; 
p15=2.0141 0 3.0148; 
tol=0.001; 
* 
p101=1.55563 0 0; 
p102=1.55563 0 2.34437; 
p103=0 0 2.34437; 
p104=2.35 0 2.3; 
* 
nr=12; 
ang=90.0; 
ang1=23.18;ang2=31.82;ang3=35.00; 
nr1=4;nr2=4;nr3=4; 
* 
p105=p101 tour ang1 p0 p6; 
p106=p105 tour (ang2 * 0.5) p0 p6; 
p107=p105 tour ang2 p0 p6; 
p108=p107 tour ang3 p0 p6; 
p108b=p108 plus p0; 
p109=0.7 0.7 0; 
p110=p108 tour 45.0 p0 p6; 
p111=p110 tour 45.0 p0 p6; 
p112=-0.7 0.7 0; 
p113=0 0.78 0; 
p113b=p113 plus p0; 
p114=-0.78 0 0; 
p115=0.78 0 0; 
p0b=p0 plus p0; 
* 
c1=p101 d 2 p1; 
c2=p1 d 6 p12 d 1 p12b d 1 p104; 
c3=p104 d 2 p102; 
c4=p102 d 8 p101; 
fluido1=daller c1 c2 c3 c4 plan; 
* 
c1=p102 d 2 p104; 
c2=p104 d 2 p14 d 3 p4 d 18 p2; 
c3=p2 d 2 p3; 
c4=p3 d 18 p5 c 3 pcv p15 c 2 pcv p102; 
fluido2=daller c1 c2 c3 c4 plan; 
fluido12=fluido1 et fluido2; 
elim tol fluido12; 
* 
oeil=0 -10000 0; 
*trac oeil fluido12; 
*opti donn 5; 
* 
fluv01=fluido12 volu nr1 rota ang1 p0 p6; 
flus01=fluido12 tour ang1 p0 p6; 
fluv02=flus01 volu nr2 rota ang2 p0 p6; 
flus02=flus01 tour ang2 p0 p6; 
fluv03=flus02 volu nr3 rota ang3 p0 p6; 
flus03=flus02 tour ang3 p0 p6; 
fluv1=fluv01 et fluv02 et fluv03; 
elim tol fluv1; 
*trac oeil cach fluv1; 
*opti donn 5; 
* 
fluv2=flus03 volu nr rota ang p0 p6; 
flus2=flus03 tour ang p0 p6; 
fluv3=flus2 volu nr rota ang p0 p6; 
flus3=flus2 tour ang p0 p6; 
fluv4=flus3 volu nr rota ang p0 p6; 
flus4=flus3 tour ang p0 p6; 
fluv=fluv1 et fluv2 et fluv3 et fluv4; 
elim tol fluv; 
*trac oeil cach fluv; 
*opti donn 5; 
* 
c1=p115 d 6 p101; 
c2=p101 c 4 p0 p105 c 2 p0 p106; 
c3=p106 d 6 p109; 
c4=p109 d 6 p115; 
bas01=daller c1 c2 c3 c4 plan; 
* 
c1=p109 d 6 p106; 
c2=p106 c 2 p0 p107 c 4 p0 p108; 
c3=p108 d 6 p113; 
c4=p113 d 6 p109; 
bas02=daller c1 c2 c3 c4 plan; 
* 
c1=p0 d 6 p115; 
c2=p115 d 6 p109; 
c3=p109 d 6 p113; 
c4=p113 d 6 p0; 

bas03=daller c1 c2 c3 c4 plan; 
* 
bas1=bas01 et bas02 et bas03; 
elim tol bas1; 
* 
*trac (0 0 1000000) bas1; 
*opti donn 5; 
* 
c1=p113b d 6 p108b; 
c2=p108b c 6 p0 p110; 
c3=p110 d 6 p112; 
c4=p112 d 6 p113b; 
bas11=daller c1 c2 c3 c4 plan; 
* 
c1=p112 d 6 p110; 
c2=p110 c 6 p0 p111; 
c3=p111 d 6 p114; 
c4=p114 d 6 p112; 
bas12=daller c1 c2 c3 c4 plan; 
* 
c1=p0b d 6 p113b; 
c2=p113b d 6 p112; 
c3=p112 d 6 p114; 
c4=p114 d 6 p0b; 
bas13=daller c1 c2 c3 c4 plan; 
* 
bas2=bas11 et bas12 et bas13; 
elim tol bas2; 
* 
*trac (0 0 1000000) (bas1 et bas2); 
*opti donn 5; 
vol1=bas1 volu 8 tran p103; 
elim tol (vol1 et fluv1); 
*trac oeil cach (vol1 et fluv1); 
*opti donn 5; 
nn=nbno vol1; 
vol1p=chan poi1 vol1; 
i=0;repe lop1 nn;i=i+1; 
  ni=vol1p poin i; 
  xi yi zi=coor ni; 
  zpstar=((2.2d0 * 2.2d0)-(xi*xi)-(yi*yi))**(0.5d0); 
  facp=(3.9d0-zpstar)/2.34437; 
  zinew=zi*facp; 
  dzi=zinew-zi; 
  depl ni plus (0 0 dzi); 
fin lop1; 
*trac oeil cach (vol1 et fluv1); 
*opti donn 5; 
vol2=bas2 volu 8 tran p103; 
*elim tol (vol2 et fluv2); 
*trac oeil cach vol2; 
*opti donn 5; 
nn=nbno vol2; 
vol2p=chan poi1 vol2; 
i=0;repe lop2 nn;i=i+1; 
  ni=vol2p poin i; 
  xi yi zi=coor ni; 
  zpstar=((2.2d0 * 2.2d0)-(xi*xi)-(yi*yi))**(0.5d0); 
  facp=(3.9d0-zpstar)/2.34437; 
  zinew=zi*facp; 
  dzi=zinew-zi; 
  depl ni plus (0 0 dzi); 
fin lop2; 
*opti donn 5; 
*trac oeil cach (vol2 et fluv2); 
*opti donn 5; 
* 
vol3=vol2 syme plan p0 p6 p111; 
vol4=vol3 syme plan p0 p6 p108; 
fluidc=fluv et vol1 et vol2 et vol3 et vol4; 
elim tol fluidc; 
*trac oeil cach fluidc; 
*opti donn 5; 
* 
nn=nbel fluidc; 
ngas=0; 
i=0;repe lop2a nn;i=i+1; 
  ei=fluidc elem i; 
  bi=bary ei; 
  xb yb zb=coor bi; 
  si (zb > 1.7); 
    ngas=ngas+1; 
    si (ngas ega 1); 
      gas=ei; 
    sinon; 
      gas=gas et ei; 
    finsi; 
  finsi; 
fin lop2a; 
* 
liqubull=diff fluidc gas; 
* 
nn=nbel liqubull; 
nbul=0; 
r2=0.6*0.6; 
i=0;repe lop2b nn;i=i+1; 
  ei=liqubull elem i; 
  bi=bary ei; 
  xb yb zb=coor bi; 
  distpcb=(xb*xb)+(yb*yb)+((zb-0.85)*(zb-0.85)); 
  si (distpcb <eg r2); 
    nbul=nbul+1; 
    si (nbul ega 1); 
      bull=ei; 
    sinon; 
      bull=bull et ei; 
    finsi; 
  finsi; 
fin lop2b; 
* 
liqu=diff liqubull bull; 
* 
fluisur=enve fluidc; 
nn=nbel fluisur; 
nstr=0; 

r2=2.2*2.2; 
i=0;repe lop3 nn;i=i+1; 
  ei=fluisur elem i; 
  bi=bary ei; 
  xb yb zb=coor bi; 
  si (zb > 3.9); 
    rb2=(xb*xb)+(yb*yb); 
    si (rb2 <eg r2); 
      nstr=nstr+1; 
      si (nstr ega 1); 
        fluidstr=ei; 
      sinon; 
        fluidstr=fluidstr et ei; 
      finsi; 
    finsi; 
  sinon; 
    distpcv=(xb*xb)+(yb*yb)+((zb-3.9)*(zb-3.9)); 
    si (distpcv <eg r2); 
      nstr=nstr+1; 
      si (nstr ega 1); 
        fluidstr=ei; 
      sinon; 
        fluidstr=fluidstr et ei; 
      finsi; 
    finsi; 
  finsi; 
fin lop3; 
*trac oeil cach fluidstr; 
*opti donn 5; 
* 
fsan=chan poi1 fluidstr; 
* 
strucq=fluidstr plus p0; 
*opti donn 5; 
struc=px4to3 strucq; 
struc=orie struc poin pcv; 
* 
*trac oeil cach fsan; 
*trac oeil cach struc; 
*opti donn 5; 
* 
nbs='NBEL' struc; 
nna=0;nnb=0;nnc=0; 
x y zp5p=coor p4; 
x y zp13p=coor p13p; 
kkk=0;repe labst nbs;kkk=kkk+1; 
  elkkk=struc 'ELEM' kkk; 
  x y z = coor ('BARY' elkkk); 
  si (< z zp5p); 
    si (ega nna 0); 
      stra=elkkk; 
      nna=nna+1; 
    sinon; 
      stra=stra et elkkk; 
    finsi; 
  sinon; 
    si (> z zp13p); 
      si (ega nnc 0); 
        strc=elkkk; 
        nnc=nnc+1; 
      sinon; 
        strc=strc et elkkk; 
      finsi; 
    sinon; 
      si (ega nnb 0); 
        strb=elkkk; 
        nnb=nnb+1; 
      sinon; 
        strb=strb et elkkk; 
      finsi; 
    finsi; 
  finsi; 
fin labst; 
* 
*trac oeil cach stra; 
*trac oeil cach strb; 
*trac oeil cach strc; 
*opti donn 5; 
* 
nbs='NBEL' struc; 
nn1=0;nn2=0;nn3=0;nn4=0; 
i=0;repe labs2 nbs;i=i+1; 
  ei=struc elem i; 
  bi=bary ei; 
  xb yb zb=coor bi; 
  si (xb > 0); 
    si (yb > 0); 
      nn1=nn1+1; 
      si (nn1 ega 1); 
        str1=ei; 
      sinon; 
        str1=str1 et ei; 
      finsi; 
    sinon; 
      nn4=nn4+1; 
      si (nn4 ega 1); 
        str4=ei; 
      sinon; 
        str4=str4 et ei; 
      finsi; 
    finsi; 
  sinon; 
    si (yb > 0); 
      nn2=nn2+1; 
      si (nn2 ega 1); 
        str2=ei; 
      sinon; 
        str2=str2 et ei; 
      finsi; 
    sinon; 
      nn3=nn3+1; 
      si (nn3 ega 1); 
        str3=ei; 
      sinon; 
        str3=str3 et ei; 
      finsi; 
    finsi; 
  finsi; 
fin labs2; 
list(nbel str1); 
list(nbel str2); 
list(nbel str3); 
list(nbel str4); 

*opti donn 5; 
* 
str1=orie str1 poin pcv; 
str2=orie str2 poin pcv; 
str3=orie str3 poin pcv; 
str4=orie str4 poin pcv; 
*vi=@visor str1 0.1 'VERT'; 
*trac (1000 -2000 3000) vi str1; 
*vi=@visor str2 0.1 'VERT'; 
*trac (1000 -2000 3000) vi str2; 
*vi=@visor str3 0.1 'VERT'; 
*trac (1000 -2000 3000) vi str3; 
*vi=@visor str4 0.1 'VERT'; 
*trac (1000 -2000 3000) vi str4; 
*opti donn 5; 
list(nbel str3); 
* 
p16l=2.1603 0.925 0;p16m=2.1603 0.925 2; 
p17l=1.3479 1.925 0;p17m=1.3479 1.925 2; 
p18l=2.95 0.925 0;p18m=2.95 0.925 2; 
p19l=2.95 1.925 0;p19m=2.95 1.925 2; 
p20l=2.95 -1.925 0;p20m=2.95 -1.925 2; 
p21l=8 -1.925 0;p21m=8 -1.925 2;p21u=8 -1.925 3.404; 
p22l=8 1.925 0;p22m=8 1.925 2; 
p23l=5 -1.925 0;p23m=5 -1.925 2;p23u=5 -1.925 3.404; 
p24l=5 1.075 0;p24m=5 1.075 2;p24u=5 1.075 3.404; 
p25l=8 1.075 0;p25m=8 1.075 2;p25u=8 1.075 3.404; 
p26l=5 1.925 0;p26m=5 1.925 2; 
c1=p16l d 4 p18l; 
c2=p18l d 4 p19l; 
c3=p19l d 4 p17l; 
c4=c 4 p17l p0 p16l; 
cam1b=daller c1 c2 c3 c4 plan; 
c1=p18l d 5 p24l d 7 p25l; 
c2=p25l d 4 p22l; 
c3=p22l d 7 p26l d 5 p19l; 
c4=p19l d 4 p18l; 
cam2b=daller c1 c2 c3 c4 plan; 
c1=p20l d 5 p23l; 
c2=p23l d 7 p24l; 
c3=p24l d 5 p18l; 
c4=p18l d 7 p20l; 
cam3b1=daller c1 c2 c3 c4 plan; 
c1=p23l d 7 p21l; 
c2=p21l d 7 p25l; 
c3=p25l d 7 p24l; 
c4=p24l d 7 p23l; 
cam3b2=daller c1 c2 c3 c4 plan; 
cam3b=cam3b1 et cam3b2; 
elim tol cam3b; 
cambb=cam1b et cam2b et cam3b; 
elim tol cambb; 
vz1=p12 'MOIN' p1; 
vz2=p12b 'MOIN' p12; 
cam1=cam1b 'VOLU' 6 'TRAN' vz1 'VOLU' 1 'TRAN' vz2; 
cam2=cam2b 'VOLU' 6 'TRAN' vz1 'VOLU' 1 'TRAN' vz2; 
cam31=cam3b1 'VOLU' 6 'TRAN' vz1 'VOLU' 1 'TRAN' vz2; 
cam32=cam3b2 'VOLU' 6 'TRAN' vz1 'VOLU' 1 'TRAN' vz2; 
cam3=cam31 et cam32; 
camb=cam1 et cam2 et cam3; 
elim tol camb; 
c1=p23m d 7 p21m; 
c2=p21m d 7 p25m; 
c3=p25m d 7 p24m; 
c4=p24m d 7 p23m; 
cam4b=daller c1 c2 c3 c4 plan; 
vz3=0 0 1.404; 
cam4=cam4b 'VOLU' 4 'TRAN' vz3; 
cam=camb et cam4; 
elim tol cam; 
* 
fluid=fluidc et cam; 
elim tol fluid; 
* 
gas=gas et cam; 
* 
surfc=enve fluid; 
fsrn=chan poi1 surfc; 
fsrn=fsrn diff fsan; 
*trac oeil fsrn; 
* 
vert=p2 moin p1; 
p0t=p0 plus vert; 
p1t=p1 plus vert; 
p17t=p17l plus vert; 
blocall=struc poin plan p0t p1t p17t tol; 
* 
tpn1 = struc poin proc p6; 
tpn2 = struc poin proc p5; 
tpn3 = struc poin proc (2.2 0 10.211111); 
tpn = tpn1 et tpn2 et tpn3; 
* 
mesh=fluid et struc et blocall et fsan et fsrn et tpn; 
tass mesh; 
* 
nn='NBEL' struc; 
coco='CHAI' 'Elements in struc =' nn; 
mess coco; 
nn='NBEL' fluid; 
coco='CHAI' 'Elements in fluid =' nn; 
mess coco; 
nn='NBNO' struc; 
coco='CHAI' 'Nodes in struc =' nn; 
mess coco; 
nn='NBNO' fluid; 
coco='CHAI' 'Nodes in fluid =' nn; 
mess coco; 
nn='NBNO' blocall; 
coco='CHAI' 'Nodes in blocall =' nn; 
mess coco; 
nn='NBNO' fsan; 
coco='CHAI' 'Nodes in fsan =' nn; 
mess coco; 
nn='NBNO' fsrn; 
coco='CHAI' 'Nodes in fsrn =' nn; 
mess coco; 
* 
opti sauv form 'vl_jrc_cavi51.msh'; 
sauv form mesh; 
opti trac psc; 
trac cach mesh; 
list(nbel str3); 
fin; 

 
The EUROPLEXUS input file reads: 
 
CAVI - 51 
$ 
 ECHO 
!CONV win 
CAST MESH 
TRID NONL ALE 
$ 
DIME 
 PT6L 1561 PT3L 9913 
 FL38 7664 COQI 3072 ZONE 2 
 NALE 521 NBLE 4639 
 nepe 160 
 ECRO 868832 
 mtti 10 
TERM 
$ 
GEOM 
  FL38 fluid 
  COQI struc 
TERM 
$ 

COMP 
  EPAI 0.13   LECT stra TERM 
       0.15   LECT strb TERM 
       0.1968 LECT strc TERM 
  COUL roug LECT bull TERM 
       bleu LECT liqu TERM 
       turq LECT gas CAM TERM 
       jaun LECT struc TERM 
$ 
GRIL LAGR LECT struc         TERM 
     EULE LECT cam fsan fsrn TERM 
     ALE  LECT fluid         TERM 
     AUTO                    AUTR 
$ 
MATE 
$ steel 
     VM23 RO 7800. YOUNG 1.6E11 NU 0.333 ELAS 1.05E8 
          TRAC 2 1.05E8 .656256E-3 1.6105E10 1.00066 
          LECT struc TERM 
     FLMP NLIQ 1 NGAS 2 
$ WT0 liquid water 

     FLUT RO 1000.0454482d0 EINT 0.0d0 GAMM 7.15d0 
          PB 3.010d9 ITER 1 ALF0 1 RREF 1000.0d0 
          CONV 0.1d0 BET0 1 KINT 0 AHGF 0 CL 0.5 
          CQ 2.56 PMIN 0 NUM 8 
          LECT liqu TERM 
$ high-pressure perfect gas (explosive bubble) 
     FLUT RO 954. EINT 3.010E6 GAMM 1.4 PB 0 
          ITER 1 ALF0 1 BET0 1 KINT 0 AHGF 0 CL 0.5 
          CQ 2.56 PMIN 0 NUM 1 
          LECT bull TERM 
$ air 
     FLUT RO 1.2 EINT 2.0833E5 GAMM 1.4 PB 0 
          ITER 1 ALF0 1 BET0 1 KINT 0 AHGF 0 CL 0.5 
          CQ 2.56 PMIN 0 NUM 1 
          LECT gas CAM TERM 
$ 
LINK COUP 
     BLOQ 123456 LECT blocall TERM 
     FSA         LECT fsan    TERM 
     FSR         LECT fsrn    TERM 
$ 
ECRI DEPL VITE ECRO TFRE 1.E-3 
          POIN LECT 1 TERM 
          ELEM LECT 1 PAS 1 10 TERM 
     FICH ALIC TFRE 5.E-4 
     FICH ALIC TEMP TFRE 5.E-5 
          POIN LECT tpn TERM 
          ELEM LECT 1 TERM 
$ 
OPTI NOTE 
 momt 2 

 log 1 
 dtml 
 REZO GAM0 0.5 
 FLMP EPS1 5.0E-4 EPS2 1.0E-4 
CALCUL TINI 0 TEND 50.E-3 
*================================================================= 
SUIT 
Post-treatment (time curves from alice temps file) 
ECHO 
* 
RESU ALIC TEMP GARD PSCR 
* 
SORT GRAP 
* 
AXTE 1000.0 'Time [ms]'  
* 
COUR  1 'dz_n1' DEPL COMP 3 NOEU LECT tpn1 TERM 
COUR  2 'dx_n2' DEPL COMP 1 NOEU LECT tpn2 TERM 
COUR  3 'dx_n3' DEPL COMP 1 NOEU LECT tpn3 TERM 
* 
trac 1 2 3 axes 1.0 'DISPL. [M]' 
trac 2 3   axes 1.0 'DISPL. [M]' 
* 
QUAL DEPL COMP 3 LECT tpn1 TERM REFE  1.23254E+0 TOLE 5.E-2 
     DEPL COMP 1 LECT tpn2 TERM REFE -1.24733E-2 TOLE 1.E-1 
     DEPL COMP 1 LECT tpn3 TERM REFE -4.01831E-2 TOLE 5.E-2 
*================================================================= 
FIN 
 

 
Some results: final pressure distribution and final water distribution: 
 

 
 
Final structure deformation and velocity distribution: 
 

 


